
Subject: Re: high-risk fertility behaviour
Posted by Liz-DHS on Fri, 23 Jan 2015 18:25:16 GMT
View Forum Message <> Reply to Message

Dear User,
Here is a response from one of our technical experts, Dr. Shea Rutstein:
Quote:There are three columns for Table 8.5: Percent of births by risk category, relative risk
category, and percent of married women by risk category.
There are single and multiple risk categories (rows of the table).  For the third column, the risks
categories are those that women would have if they became pregnant.  Therefore, women who
have no children and are between 17 years, 3 months of age and 34 years, 2 months of age
would be those for question 1 (risk category 2).
Question 2:  For the first two columns, the preceding birth interval is standard recode variable b11
<24 if there were 2 or more births (months of the preceding  birth interval--twins, triplets have the
same value to the birth of a preceding pregnancy).  For the third column, the preceding birth
interval is standard recode variable v222<15 if there were 1 or more births (months between last
birth and interview).

Here is the CSPRO code for Table 8.5
for ch in REC21_EDT do
    months = V008 - B3(ch);
    if months < 60 then                          { For all children born  }
      agemoth = int( (B3-V011)/12 );             { in the last five years }
      agemo18 = ( agemoth < 18 );
      agemo34 = ( agemoth > 34 );
      inter24 = 0;
      if !special(B11) then
        inter24 = ( B11 < 24 );
      endif;
      { adjust multiple births to give the same order as that of the first of the multiples }
      xorder = BORD;
      if B0 > 1 then xorder = BORD - B0 + 1 endif;
      bord3 = ( xorder > 3 );
      frstb = ( xorder = 1 );
      box agemo18 : agemo34 : inter24 : bord3 : frstb => rowt805;
              0   :    0    :    0    :   0   :   0   => 1;  { no risk }
              0   :    0    :    0    :   0   :   1   => 2;  { unavoidable risk }
              1   :    0    :    0    :   0   :       => 3;  { single risk }
              0   :    1    :    0    :   0   :       => 4;
              0   :    0    :    1    :   0   :       => 5;
              0   :    0    :    0    :   1   :       => 6;
              1   :    0    :    1    :   0   :       => 7;  { multiple risk }
              1   :    0    :    0    :   1   :       => 7;
              1   :    0    :    1    :   1   :       => 7;
              0   :    1    :    1    :   0   :       => 8;
              0   :    1    :    0    :   1   :       => 9;
              0   :    1    :    1    :   1   :       => 10;
              0   :    0    :    1    :   1   :       => 11;

                  :         :         :       :       => default;
      endbox;
      box rowt805 => norisk;
              1   => rowt805;
                  => notappl;
      endbox;
      box rowt805 => unavoid;
              2   => rowt805 - 1;
                  => notappl;
      endbox;
      box rowt805 => singrisk;
             3-6  => rowt805 - 2;
                  => notappl;
      endbox;
      box rowt805 => multrisk;
             7-11 => rowt805 - 6;
                  => notappl;
      endbox;
      currmar1 = notappl;
      colt805  = 1;
      xtab( t805, rweight );
      if B5 <> 1 then                { For dead children  }
        colt805 = 2;
        xtab( t805, rweight );
      endif;
    endif;
  enddo;

  if V502 = 1 then                         { For currently married women }
    colt805  = notappl;
    currmar1 = 1;
    agemoth = V008 - V011;
    agemo18 = ( agemoth < 17*12+3 );
    agemo34 = ( agemoth > 34*12+2 );
    inter24 = 0;
    if V201 > 0 then
       inter24 = ( V222 < 15 );
    endif;
    bord3 = ( V201 > 2 );
    frstb = ( V201 = 0 );
    box agemo18 : agemo34 : inter24 : bord3 : frstb => rowt805;
            0   :    0    :    0    :   0   :   0   => 1;  { no risk }
            0   :    0    :    0    :   0   :   1   => 2;  { unavoidable risk }
            1   :    0    :    0    :   0   :       => 3;  { single risk }
            0   :    1    :    0    :   0   :       => 4;
            0   :    0    :    1    :   0   :       => 5;
            0   :    0    :    0    :   1   :       => 6;
            1   :    0    :    1    :   0   :       => 7;  { multiple risk }

            1   :    0    :    0    :   1   :       => 7;
            1   :    0    :    1    :   1   :       => 7;
            0   :    1    :    1    :   0   :       => 8;
            0   :    1    :    0    :   1   :       => 9;
            0   :    1    :    1    :   1   :       => 10;
            0   :    0    :    1    :   1   :       => 11;
                :         :         :       :       => default;
    endbox;
    if V312 = 6 then rowt805 = 1 endif;
    box rowt805 => norisk;
            1   => rowt805;
                => notappl;
    endbox;
    box rowt805 => unavoid;
            2   => rowt805 - 1;
                => notappl;
    endbox;
    box rowt805 => singrisk;
           3-6  => rowt805 - 2;
                => notappl;
    endbox;
    box rowt805 => multrisk;
           7-11 => rowt805 - 6;
                => notappl;
    endbox;
    xtab( t805, rweight );
  endif;

